.trip layout
All integers (unsigned int in C) are MSB (Most Significant Bit)
There is 1 header record, followed by a number of item records
Header record
4 bytes		constant ‘TRPL’
4 bytes int	File size ’00 00 b9 52’ = 47442. (This is excluding the header length)
1 byte int	0A (Unknown, version ?)
4 byte int	’00 00 00 14’ = 20 Nr of items to follow.
1 byte char	‘09’ Terminator char. This is also used for all items.
[image: ]


Item Records
An Item record has a different size for each datatype, but they all share the same structure.
4 byte int	’00 00 00 15’ = 21 Size of the key name
N bytes		‘mPreserveTrackToRoute’ Keyname. 
4 byte int	’00 00 00 02’ Size of the value. (including the terminator)
	1
	Byte Integer

	3
	Dword integer 

	4
	Dword integer

	3 and 4 are both Dword integer, maybe one is signed, and the other unsigned?

	7
	Bool

	14
	String. Strings are 32 bit UTF

	128
	List type. Looks like it contains a list of items. This type I did not investigate.
mLocation: via/shaping points?
mAllRoutes: ghost points?


1 byte int	’07’ DataType







N bytes		‘00’	Boolean value for False
1 byte		’09’ Terminator char
[image: ]


Sample of a string value:
’00 00 00 09’ 	Length of keyname
‘mFilename’	Keyname
’00 00 00 83’	Length of value
‘0E’		Datatype
’00 80’		Length of string
’30 00 00 00’	first char ‘0’ until ’70 00 00 00’ ‘p’
‘09’		Terminator char[image: ]

Sample of a list value
’00 00 00 0A’ 	length of keyname
‘mLocations’	Keyname
’00 00 04 89’	= 1161 Length of value
‘80’		=128 datatype
This list appears to have ‘LCTN’, ‘mAttr’, ‘mIsDFSpoint’ etc. as sub items. Did not look further.

[image: ]

Here’s some Delphi Pascal code that demonstrates how the data can be read/modified. You can use it for documentation purposes.
procedure Swap32(var X);
asm
   mov ecx, [eax]
   bswap ecx
   mov [eax], ecx
end;

procedure TFrmTrip.ReadFile(Sender: TObject);
var Item: integer;
    Indx: integer;
    KeyLen: DWORD;
    KeyName: PAnsiChar;
    ValueLen: Dword;
    DataType: byte;
    ValueData: array of byte;
    BytesRead: integer;
    ValueBool: Boolean;
    ValueByte: byte;
    ValueDword: DWORD;

    ValueString: string;
    PCharLen: smallint;
    PCharDword: DWORD;
    Terminator: byte;
    SaveFilePos: int64;
    Msg: string;

const BoolValues: array[False .. True] of string = ('False', 'True');

begin
  memo1.Lines.Clear;

  if not OpenDialog1.Execute then
    exit;
  FileMode := fmOpenReadWrite;
  AssignFile(F, OpenDialog1.FileName);
  Reset(F,1);
  BlockRead(F, HeaderRec, SizeOf(HeaderRec));
  Swap32(HeaderRec.FileSize);
  Swap32(HeaderRec.Items);
  Msg := Format('Items expected %d',[HeaderRec.Items]);
  Memo1.Lines.Add(Msg);

  Item := 0;
  while (true) do
  begin
    inc(Item);
    BlockRead(F, KeyLen, SizeOf(KeyLen), BytesRead);
    if (BytesRead = 0) then
      break;
    Swap32(KeyLen);
    KeyName := AnsiStrAlloc(KeyLen +1);
    FillChar(KeyName^, KeyLen+1, 0);
    BlockRead(F, KeyName^, KeyLen, BytesRead);
    if (BytesRead = 0) then
      break;
    BlockRead(F, ValueLen, SizeOf(ValueLen), BytesRead);
    if (BytesRead = 0) then
      break;
    Swap32(ValueLen);
    BlockRead(F, DataType, SizeOf(DataType), BytesRead);
    if (BytesRead = 0) then
      break;

    SaveFilePos := FilePos(F);

    Msg := format('%d : %s =>', [Item, StrPas(KeyName)]);
    case DataType of
       1: begin // Byte integer;
            BlockRead(F, ValueByte, SizeOf(ValueByte), BytesRead);
            if (BytesRead = 0) then
              break;
            Msg := Msg + Format('%d', [ValueByte]);
          end;
       3,4:begin // DWORD integer;
            BlockRead(F, ValueDword, SizeOf(ValueDword), BytesRead);
            if (BytesRead = 0) then
              break;
            Swap32(ValueDWord);
            Msg := Msg + Format('%d', [ValueDWord]);
          end;
       7: begin // Bool
            BlockRead(F, ValueBool, SizeOf(ValueBool), BytesRead);
            if (BytesRead = 0) then
              break;
            if (StrPas(KeyName) = 'mImported') and
               (Tbutton(Sender).Tag > -1) and
               (Tbutton(Sender).Tag < 2) then
            begin
              Seek(F, SaveFilePos);
              ValueBool := boolean(Tbutton(Sender).Tag);
              BlockWrite(F, ValueBool, SizeOf(ValueBool));
            end;
            Msg := Msg + Format('%s', [BoolValues[ValueBool]]);
          end;
       14:begin // String
            BlockRead(F, PCharLen, SizeOf(PCharLen), BytesRead);
            if (BytesRead = 0) then
              break;
            PCharLen := (Swap(PCharLen) div 4) -1;
            ValueString := '';
            for Indx := 0 to PcharLen do
            begin
              BlockRead(F, PCharDword, SizeOf(PCharDword), BytesRead);
              if (BytesRead = 0) then
                break;
              ValueString := ValueString + char(PCharDword);
            end;
            Msg := Msg + Format('%s', [ValueString]);
          end;
       else
          begin
            Msg := Msg + Format('Datatype %d, Len: %d', [datatype, ValueLen]);
          end;
    end;
    StrDispose(KeyName);
    Memo1.Lines.Add(Msg);

    BlockRead(F, Terminator, SizeOf(Terminator), BytesRead);
    if (BytesRead = 0) then
      break;

//Failsafe, if we have missed some data
    Seek(F, SaveFilePos + ValueLen);
  end;

  CloseFile(F);
end;


procedure TFrmTrip.Button1Click(Sender: TObject);
begin
  ReadFile(Sender);
end;

end.

